Algorithme compte gouttes pour les décimales de e

1 Le principe

La fonction exponentielle peut se développer en série entière (somme des termes d'une suite) par :

$$e^x = 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \frac{x^4}{4!} \cdots + \frac{x^n}{n!} + \ldots$$

où $n!$, factorielle n, vaut : $n! = n \times (n-1) \times (n-2) \ldots 3 \times 2 \times 1$

Pour calculer la constante e, il suffit alors de prendre $x = 1$. On obtient alors :

$$e = 1 + 1 + \frac{1}{2!} + \frac{1}{3!} + \frac{1}{4!} \cdots + \frac{1}{n!} + \ldots$$

Que l'on peut écrire avec la formule de Horner

$$e = 2 + \frac{1}{2}\left(1 + \frac{1}{3}\left(1 + \frac{1}{4}\left(1 + \frac{1}{5}\left(1 + \frac{1}{6}\left(1 + \frac{1}{7}(\ldots)\right)\right)\right)\right)\right)$$

Les facteurs $\left(\frac{1}{2}, \frac{1}{3}, \frac{1}{4}, \frac{1}{5}, \ldots, \frac{1}{n}, \ldots\right)$ peuvent être considérés comme une base à pas variable. Dans cette base un peu particulière, l'expression de e est $[2, 1, 1, 1, \ldots]$.

On remarque alors que dans cette base, e devient alors un nombre d'une simplicité stupéfiante.

Il ne reste donc qu'à trouver un algorithme qui permette de transcrire $[2, 1, 1, 1, \ldots]$ dans notre système à base 10. On applique alors l'algorithme de Horner (cf première) avec les fractions $\left(\frac{1}{2}, \frac{1}{3}, \frac{1}{4}, \frac{1}{5}, \ldots, \frac{1}{n}, \ldots\right)$

2 L'algorithme

2.1 Nombre de digits

Tout d'abord, un petit calcul en ce qui concerne l'encombrement mémoire. Dans la forme de Horner, hormis le premier terme, on voit que le pas $\frac{1}{n}$ de la base est inférieur ou égal à $\frac{1}{3}$. La valeur exacte $\frac{1}{3}$ reviendrait à considérer la base 3 . Pour avoir un chiffre significatif dans notre système décimal pour un nombre écrit en base avec x "*digit*", on a

$$3^x = 10 \quad \Leftrightarrow \quad x = \frac{\ln 10}{\ln 3} \approx 2,20.$$

Comme le pas diminue au cours du développement, pour avoir une décimale de e, on aura besoin au maximum de 2 "*digit*".

Remarque : À partir du 9e terme, un seul "*digit*" sera même nécessaire.

2.2 Tableau donnant 4 chiffres significatifs

A	e	1	1	1	1	**1**	1	1	1
B	=	1	2	3	4	**5**	6	7	8
initialisation		2	1	1	1	**1**	1	1	1
Premier chiffre									
x 10		20	10	10	10	**10**	10	10	10
retenue		7	4	3	2	**1**	1	1	0
somme	2	27	14	13	12	**11**	11	11	10
reste		7	0	1	0	**1**	5	4	2
Deuxième chiffre									
x 10		70	0	10	0	**10**	50	40	20
retenue		1	3	0	3	**9**	6	2	0
somme	7	71	3	10	3	**19**	56	42	20
reste		1	1	1	3	**4**	2	0	4
Troisième chiffre									
x 10		10	10	10	30	**40**	20	0	40
retenue		8	6	9	8	**3**	0	5	0
somme	1	18	16	19	38	**43**	20	5	40
reste		8	0	1	2	**3**	2	5	0
Quatrième chiffre									
x 10		80	0	10	20	**30**	20	50	0
retenue		2	5	6	6	**4**	7	0	0
somme	8	82	5	16	26	**34**	27	50	0
reste		2	1	1	2	**4**	3	1	0

On obtient alors : $e \approx 2,718$

Les deux premières lignes (A et B) sont les numérateurs et dénominateurs des pas de la base à pas variable.
La troisième ligne (initialisation), correspond à l'expression de π dans cette base.
On remplit la dernière colonne des lignes "*retenue*" par des 0.

L'algorithme de conversion s'effectue de droite à gauche par groupe de 4 lignes : ($\times 10$), retenue, somme et reste.
Le principe consiste à multiplier le digit par 10, à additionner la retenue, à diviser la somme par le chiffre de la colonne B qui donne d'une part le quotient (la retenue de la colonne suivante) et d'autre part le reste.

Remplissage de la colonne en rouge (colonne 5) :

- On remplit la ligne $\times 10$ en multipliant la ligne précédente par 10 : $1 \times 10 = 10$.
- On additionne ce nombre à la retenue que l'on a calculé avec la colonne précédente : $10 + 1 = 11$

- On effectue la division euclidienne de la somme par le nombre de la ligne B de la même colonne : $11 = 5 \times 2 + 1$
- On place le reste 1 dans la ligne reste.
- On place le quotient dans la ligne retenue de la colonne suivante : 2

On réitère le procédé sur toutes les colonnes des 4 lignes du premier chiffre et l'on obtient 27 comme dernière somme. On divise par 10, on prend 2 comme premier chiffre de e et 7 comme reste.
On réitère de nouveau ce procédé aux 4 lignes suivantes : on obtient le deuxième chiffre : 7 et ainsi de suite pour obtenir les chiffres suivants : 1 et 8

2.3 Programme

Voici un programme que vous pouvez entrer dans votre calculette.

Pour déterminer le nombre de tableaux de 4 lignes que l'on doit effectuer, on fera l'approximation que 2 "*digits*" donne un chiffre significatif. C'est pour cette raison fera Ent($N/2$) boucles correspondant aux tableaux. Pour connaître 10 chiffres significatifs, on rentrera $N = 20$.

N :	Nombre de *digits*
I :	Compteur
J :	Rang de la décimale calculée
B :	Valeur du pas et de la colonne
R :	Retenue
E :	Dernier reste du tableau précédent (initialisée à 0)
F :	Valeur de la décimale calculée

- Lignes 7, 8, 9, 10 : on rentre 2 comme 1er nombre dans la liste L_1 puis $N-1$ fois le chiffre 1.
- Ligne 11 : on se positionne sur la colonne la plus à droite.
- Ligne 14 : on calcule la somme de 10 fois le reste $L_1(B)$ du tableau précédent et de la retenue.
- Ligne 15 : on calcule le reste.
- Ligne 16 : on calcule le quotient.
- Ligne 17 on prend la colonne suivante vers la gauche.
- Ligne 19 : on calcule un chiffre significatif avec le reste précédent de la colonne reste et la retenue divisée par 10.
- Ligne 20 : Pour assembler le chiffre significatif (sortie "Output" sur la Ti), on positionne les chiffres sur la 4e ligne par exemple, à la colonne J.
- Lignes 21, 22, 23 après le premier chiffre (2), on met une virgule.
- Ligne 25 : on calcule le reste de la colonne 1.

```
NORMAL FLOTT AUTO RÉEL RAD MP
PROGRAM:COMPTGE
:EffÉcran
:EffListe L1
:Prompt N
:0→E
:1→J
:2→L1(1)
:For(I,2,N)
:1→L1(I)
:End
:For(I,1,partEnt(N/2))
:N→B
:0→R
:While B>0
:R+10L1(B)→R
:R-BpartEnt(R/B)→L1(B)
:partEnt(R/B)→R
:B-1→B
:End
:E+partEnt(R/10)→F
:Output(4,J,F)
:If J=1
:Then
:Output(4,2,",")
:J+1→J
:End
:R-10partEnt(R/10)→E
:J+1→J
:End
:Output(6,1,"e INTEGRE ")
:Output(6,11,2.718281828)
```

Variables : N, I, J, B, R, E, F entiers
L_1 liste

Entrées et initialisation
- Effacer écran
- Effacer liste L_1
- Lire N
- $0 \to E$
- $1 \to J$
- $2 \to L_1(1)$
- **pour** I de 2 à N **faire**
  - $1 \to L_1(I)$
- **fin**

Traitement
- **pour** I de 1 à Ent$(N/2)$ **faire**
  - $N \to B$
  - $0 \to R$
  - **tant que** $B > 0$ **faire**
    - $R + 10L_1(B) \to R$
    - $R - B\text{Ent}\left(\dfrac{R}{B}\right) \to L_1(B)$
    - $\text{Ent}\left(\dfrac{R}{B}\right) \to R$
    - $B - 1 \to B$
  - **fin**
  - $E + \text{Ent}\left(\dfrac{R}{10}\right) \to F$
  - Placer F à la 4e ligne de la Je colonne
  - **si** $J = 1$ **alors**
    - Placer "," à la 4e ligne de la 2e colonne
    - $J + 1 \to J$
  - **fin**
  - $R - 10\text{Ent}\left(\dfrac{R}{10}\right) \to E$
  - $J + 1 \to J$
  - Afficher sur la 6e ligne "e intégré 2,718281828"
- **fin**

Avec une Ti 82 ou 83, on obtient pour $N = 40$ à comparer à la valeur intégrée dans la calculatrice (19 décimales)

On trouve :

$e \approx 2{,}718\,281\,828\,459\,045\,235\,3$

```
NORMAL FLOTT AUTO RÉEL RAD MP
N=?40
                                   Fait
........................................
2,7182818284590452353
e INTEGRE 2.718281828
```